```
> # This week, we're going over the following commands:
> # plot, solve, VolumeOfRevolution, and FunctionAverage
> # Let's start with plot; from the examples in the documentation, we can see that the command is
    fairly similar to the int and RiemannSum commands from last week
> # That is, we need to specify a function alongside the independent variable (with the option for an
    interval for the variable)
> plot(x^2, x = 0 ..1)
```




```
> plot(x^2, x)
```




```
> # To graph multiple functions at the same time, we need to surround the functions with [ ].
> # Essentially, we are creating a list (or array) of functions to display
> plot([x^2, x^3, exp(x), sin(x)], x = -Pi/2 .. Pi/2)
```




```
> # Finally, we also need to include a legend for Q1.
> # This is done by creating a list of labels for the functions
> f := x→cos(x)
```

$$f := x \mapsto \cos(x) \quad \textbf{(1)}$$

```
> g := x→sin(x)
```

$$g := x \mapsto \sin(x) \quad \textbf{(2)}$$

```
> plot([f(x), g(x)], x=-Pi..Pi, legend=[f(x), g(x)])
```




```
> # If we want to specify some specific text for the legend, we need to use strings
> # All we need to do is put the text between quotes
> plot( [ f(x), g(x) ], x =-Pi ..Pi, legend= ["f(x)", "g(x)"])
```




```
>
>
> # Next, we'll take a look at the solve command which lets us solve systems of equations for some
    independent variables
> # All we need for the assignment is to cover a single equation with a single variable, so we'll look
    at that first
> # Like before, we need to specify an algebraic expression alongside the independent variable we
    want to solve for
> solve(f(x) = g(x), x)
```

$$\frac{\pi}{4} \qquad (3)$$

```
> # Of course, it's not realistic for Maple to return all solutions (since there are infinitely many to
    the equation we used above); however, we must pay attention to the fact that Maple missed
    solutions that may appear obvious to us (like in this case)
> # As a result, the only conclusion we can make using the solve command is that the returned
    numbers are solutions, but not that they are the only solutions
> # [Optional] To solve systems of equations with multiple variables, we need to specify a list of
    equations, and a list of variables like we did with the plot command
> h := (x, y) → x^2 - y^2
```

$$h := (x, y) \mapsto x^2 - y^2 \quad (4)$$

```
> g := x→x^3
```

$$g := x \mapsto x^3 \quad (5)$$

```
> solve([h(x,y)=0, g(x)=y], [x,y])
```

$$[[x = 0, y = 0], [x = 1, y = 1], [x = -1, y = -1], [x = RootOf(\_Z^2 + 1), y = -RootOf(\_Z^2 + 1)]] \quad (6)$$

```
> # Sometimes, if the equations don't have nice solutions, Maple will use the notation "RootOf"
> # You probably won't need to worry about this, but it just refers to the root of a polynomial - in
    this case, z^2 + 1 = 0 for x and -x for y. That is, x = i and y =
    -i where i respresents the imaginary unit(square root of -1)
> # Other times, Maple might not find any solutions
> solve([f(x)=g(x), x<0], [x])
```

$$[\,] \quad (7)$$

```
> # Again, this doesn't mean that the system of equations doesn't have any solutions; just that maple
    couldn't find any.
>
>
> # Before we move onto the last 2 commands, let's quickly look at the restart command
> # It can happen that we have done a lot of work in a particular worksheet and we have a large
    number of variables
> # If we want to refer to different expressions with names that we have used before, we might run
    into problems if we aren't careful
> # for example, Q3 and Q4 both refer to a function called f. If we aren't careful and forget to
    overwrite Q3's f(x), we will get a wrong answer for Q4
> # The problem is that Maple will happily find the correct value for Q3's function which means we
    might miss the mistake
> # To prevent this, we can use the restart command that clears all variables from memory -
    essentially, think of it as a separate worksheet
> # This also means that if we use restart at some point in our worksheet, we will have to rerun
    everything in order if we make changes to the worksheet before the restart command
> # We can do this Evaluate > Execute Worksheet at the top(Ctrl + Shift + Enter on Windows)
> solve(f(x)=0, x)
```

$$\frac{\pi}{2} \quad (8)$$

```
> restart
> solve(f(x)=0, x)
```

$$RootOf(f(\_Z)) \quad (9)$$

```
> # Here, we can clearly see that something went wrong because Maple didn't even attempt to plug
    x into f(x)
> # Generally, it is good practice to use the restart command at the start of a new question (e.g. at
    the start of Q1, ..., Q4)
>
>
> # For the VolumeOfRevolution and FunctionAverage commands, we need to import the
    Caclulus1 package  (similar to RiemannSum)
```

```
> with(Student[Calculus1]) :
>
>
> # Now let's look at VolumeOfRevolution
> # From the documentation, we can see that it works essentially in the same way as the commands
    before with a function and interval
> VolumeOfRevolution(sin(x), x = 0 ..Pi)
```

$$\frac{\pi^2}{2} \tag{10}$$

```
> # Similar to RiemannSum, there is an output option that you need to know about for the
    assignment
> # The options you need to know are:
> # output = value, plot, or integral
> # output = value is the same as not specifying the output at all (i.e. it's the default)
> VolumeOfRevolution(sin(x), x = 0 ..Pi, output = value)
```

$$\frac{\pi^2}{2} \tag{11}$$

```
> # output = plot displays a plot (as the name suggests) - Notice that the resulting 3D plot is
    interactive
> VolumeOfRevolution(sin(x), x = 0 ..Pi, output = plot)
```



The solid of revolution created on $0 \le x \le \pi$ by rotation of $f(x) = \sin(x)$ about the axis $y = 0$.

```
> # Finally, output = integral displays the integral that evaluates to the volume of the resulting
    object
> VolumeOfRevolution(sin(x), x = 0..Pi, output = integral)
```

$$\int_0^{\pi} \pi \sin(x)^2 \, dx \qquad \textbf{(12)}$$

```
>
>
> # Finally, let's look at the FunctionAverage command
> # Again, we need a function and a variable:
> FunctionAverage(sin(x), x = 0..Pi)
```

$$\frac{2}{\pi} \qquad \textbf{(13)}$$

```
> evalf(solve(sin(x) = 2/Pi, x), 4)
```

$$0.6901 \qquad \textbf{(14)}$$

```
> plot([sin(x), 2/Pi], x = 0..Pi)
```


1
0.8
0.6
0.4
0.2
0
π/8
π/4
3π/8
π/2
5π/8
3π/4
7π/8
π
x